============Threshold_Sampling==========================





Now Asynchronous sampling means recording a data point only when something happens. Consider a 3Bit threshold detecting comparator. In the example above, a sine wave sweeps over a frequency range. An AADC will change state 16 times for every sine wave. Say at each transition, 13bits representing the period between transitions, together with 3bits worth of value are sent out. So around 60X16bit data points get sent out.



The receiver only receive 60 time points with three extra bits added to include a value. Reconstructing the original signal for the moment looks a little lacking. But extrapolation can change all that.

=======Spice_Code===============================

```
Level_Crossing_Freq_Sweep
*
*       VIN
*
*      _|_  ________________ THO  __|_   _____________ TH1  __|_   |  THO
*     /   \            |         /    \  |          |      /    \  |
*    //\   \           |  |\    -\____/  |      |\   -\____/    | |\
*    \    \//         _|  | \_   |       |_     | \_    |       _| | \_
*     \___/            |  | /    |  OFFO   |    | /   OFF1       | |  /   ______   DAC    \__|VOUTA|
*      _|_               |/      |         |    |/              |/   |          |        /   |_____|
*      ///                      |DO|            |D1|               |D2|__|   ___|______/
*                               |__|            |__|               |__|  |  |
*                                 |               |__________________________|
*                                 |_______________________________________|
*
*
*
*=========Create_Signal=================
Vtime      Vtime  0      DC      0      PWL( 0 0 1 1)
BVIN       VIN    0      V =           .5*sin(27000*(V(Vtime))^3) +.5


BTHO       DO     0      V =     u( V(VIN)  -1/2)
BOFFO      VIN    OFFO   V =     V(DO)/2
BTH1       D1     0      V =     u( V(OFFO) -1/4)
BOFF1      OFFO   OFF1   V =     V(D1)/4
BTH2       D2     0      V =     u( V(OFF1) -1/8)
BOFF2      OFF1   OFF2   V =     V(D2)/8
BTH3       D3     0      V =     u( V(OFF2) -1/16)
BOFF3      OFF2   OFF3   V =     V(D3)/16
BTH4       D4     0      V =     u( V(OFF3) -1/32)
BDAC       VOUTA  0      V =     (V(DO)/2+V(D1)/4+V(D2)/8 )

Btrig      trig   0      V =     1*u( V(VOUTA)- V(VIN) +5m )
CHP        VOUTA  HP     .1n
```

```
RHP       HP     0      100K
BTRIG2    trig2  0      V = u( 100*abs(V(HP))  -.1   )


XTOGLE    trig2   D0B    TOGLE
XTOGLE0   D0B     D1B    TOGLE
XTOGLE1   D1B     D2B    TOGLE
XTOGLE2   D2B     D3B    TOGLE
XTOGLE3   D3B     D4B    TOGLE
XTOGLE4   D4B     D5B    TOGLE
XTOGLE5   D5B     D6B    TOGLE
BDAC2     VOUTB   0      V =    128-(V(D6B)*64+V(D5B)*32+V(D4B)*16+V(D3B)*8+V(D2B)*4 +V(D1B)*2 +V(D0B))

XPOS_E2   trig2   PEOUT  POS_E2
XS_H1     Vtime   trig   VtimeS      SH

.control
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      2u     100m      0      2u
set       pensize = 2

*plot     vin  vouta  voutb/50  xlimit  1m  100m
plot      vin  vouta            xlimit  1m  100m


let  numb = length(vtimes)-101
unlet index
unlet val
unlet index
let index = vector(62)
let val   = vector(62)*0 +.5
let tref  = vector(62)*0 +.1
let dt    = vector(62)*0

let i = 100
repeat  $&numb
let tt = peout[i]

if (tt > .1)
let k = voutb[i]
let index[k] = k
let value = vin[i]
let val[k] = value
let tref[k] = vtimes[i]
end

let i = i+1
end

let tref[0] = 0
let dt    = vector(62)*0

let m = 1
repeat 60
let dt[m] = tref[m] -tref[m-1]
let m = m+1
end

plot val dt*50  vs index
plot tref  vs index
plot val vs tref

.endc

.MODEL    SW      SW(    VT=.5 VH=.1  RON=1 ROFF=100MEG)
.MODEL    DD2      D( IS=3.15e-18 RS=1  CJ0=2f)

*=========POS_Edge2============================================
*
*         |\             VBF  ____              ___
*  |IN |__|3\_________________|    \           |   \
*  |___|  | /  |          VLP| PE  \_|OUT|  |IN |_|PE  \_|OUT|
*         |/   |/\  /\  /\____|    /  |___|  |___| |    / |___|
*                \/  \/   |   |___/              |___/
*               RLP      _|_
*                        ___
*                   CLP   |
*                        _|_
*                        ///
*
.SUBCKT   POS_E2  IN    OUT
 BBUF     VBF    0      V =   2*u( v(IN )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      .1n   IC=0
 BAND     OUT    0      V =   u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
.ENDS     POS_E2

*=========Sample_Hold============================================
*
*             IN2
*     ___   |\    /         IN3 |\     ___
*    |IN |_|1\__/  ____________|2\____|OUT|
*    |___|  | / S1   _|_        | /    |___|
*           |/       ___        |/
*                C1   |
*                    _|_
*                    ///
.SUBCKT   SH      IN     CNTL      OUT
 B1       IN2     0      V =       v(IN )
 S1       IN2     IN3    CNTL     0    SW
 C1       IN3     0      .1u
```

```
 R1       IN3    0      100Meg
 B2       OUT    0      V =       v(IN3 )
.ENDS     SH

*=========POS_Edge===========================================
*
*  ___   |\             VBF ____                  ___
* |IN |__|3\_______________|    \   ____   ____  |   \   ____
* |___|  | /  |          VLP|  PE \_|OUT|  |IN |_| PE \_|OUT|
*        |/  |/\  /\  /\____|     / |___|  |___| |    / |___|
*               \/  \/   |  |___/                |__/
*              RLP      _|_
*                       ___
*                  CLP   |
*                       _|_
*                       ///
*
.SUBCKT   POS_E  IN     OUT
 BBUF     VBF    0      V =   u( v(IN )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      .1n   IC=0
 BAND     OUT    0      V =    u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
.ENDS     POS_E

*=========TOGGLE=========================================
*       _____________________________________
*      |                                     |
*      |   /    VC1 |\  _NVC1 /    VC2  |\   |  ___
*      |__/  _______| \/ \___/  ________| \__|_|OUT|
*        S1    _|_  | /\_/  S2    _|_   | /    |___|
*              ___  |/            ___   |/
*           C1  |              C2  |
*               |                  |
*              ///                ///          |\
*  ___   |\             VBF ____              DPE   __|B\__\S2
* |IN |__|3\_______________|    \ PE               |  | /  /
* |___|  | /  |          VLP|  PE \___/\  /\  /\_____|  |/
*        |/  |/\  /\  /\____|     /     \/  \/   |   |  |\
*               \/  \/   |  |___/      RLP2     _|_  |__|M\__\S1
*              RLP1     _|_                 CLP2___     | /  /
*                       ___                      |      |/
*                  CLP1  |                      _|_
*                       _|_                     ///
*                       ///
.SUBCKT   TOGLE   IN     OUT2
 BBUF     VBF     0      V =    u( v(IN )-.5 )
 RLP      VBF     VLP    1k
 CLP      VLP     0      5n
 BAND     VPE     0      V =    u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.5)
 RLP2     VPE     VLP2   .3k
 CLP2     VLP2    0      2n
 BBRK     BRK     0      V =     1-u(v(VLP2 )-.2)
 BMAK     MAK     0      V =     u(v(VLP2 )-.9)
 S1       OUT     VC1    MAK     0        SW
 S2       NVC1    VC2    BRK     0        SW
 C1       VC1     0      10n
 C2       VC2     0      10n
 R1       VC1     0      10Meg
 R2       VC2     0      10Meg
 BINV     NVC1    0      V =     1-u(v(VC1)-.2)
 BOUT     OUT     0      V =     u(v(VC2)-.2)
 BOUT2    OUT2    0      V =     1- u(v(VC2)-.2)
.ENDS     TOGLE

.end
```


8.18.11_3.11PM
dsauersanjose@aol.com
Don Sauer